
agua_sat() 
Prgm 

Local 
m_ea,m_dd,l_ingad,error_d,error_n,opc_1,tp_req,tp_uni,tp_lim,tp_s,opc_2,x
_s,v_s,u_s,h_s,s_s,tp,cont,i,l_ingadi,l_ingads,ingad_s,ingad,x,v,u,h,s,tp
_unio2,tp_uni1,tp_conv,tpa89_ip,rta_1,rta,y_f,y_g 
If getType(termo\tabla_a8)≠"MAT" or getType(termo\tabla_a9)≠"MAT" or 
getType(termo\tp_a89)≠"FUNC" Then 
 Pause "Error: Deben estar las matrices 'termo\tabla_a8' y 
'termo\tabla_a9', y la función 'termo\tp_a89'. Envíelas a la calculadora 
e intente de nuevo" 
 Stop 
EndIf 
setMode("Exact/Approx","APPROXIMATE")→m_ea 
setMode("Display Digits","FLOAT 6")→m_dd 
{"x","v","u","h","s"}→l_ingad 
"Error: Revise el valor ingresado de "→error_d 
"Error: Se debe(n) ingresar únicamente número(s) ["→error_n 
ClrIO 
Loop 
 Dialog 
  Title "AGUA SATURADA v1.0" 
  Text "Propiedades del Agua Saturada: " 
  Text " Tabla de Temperatura (A.8)" 
  Text " Tabla de Presión     (A.9)" 
  Text "" 
  DropDown " Ingresar:",{"Temperatura","Presión"},opc_1 
  Text "" 
  Text "" 
  Text "  Por: Froilán A. Morales S." 
  Text "   <froilan1986@hotmail.com>" 
  Text "   UIS, Ingeniería Mecánica." 
 EndDlog 
 If ok=0 
  Exit 
 If opc_1=1 Then 
  "T"→tp_req 
  {"°C","K","°F","R"}→tp_uni 
  {".01","374.15"}→tp_lim 
 Else 
  "p"→tp_req 



  {"bar","Pa","atm","mmHg","mmH2O","psi","inHg","inH2O"}→tp_uni 
  {".01","221.29"}→tp_lim 
 EndIf 
 tp_uni[1]→tp_uni1 
 Loop 
  Dialog 
   Title tp_lim[1]&tp_uni1&"≤"&tp_req&"≤"&tp_lim[2]&tp_uni1 
   Request tp_req,tp_s,0 
   DropDown "  Unidad:",tp_uni,opc_2 
   Text "Ingresos adicionales (uno opcional): " 
   Request "x",x_s 
   Request "v,m³/kg",v_s 
   Request "u,kJ/kg",u_s 
   Request "h,kJ/kg",h_s 
   Request "s,kJ/(kg*K)",s_s 
  EndDlog 
  If ok=0 
   Exit 
  Try 
   expr(tp_s)→tp 
  Else 
   Pause error_d&tp_req 
   Cycle 
  EndTry 
  If getType(tp)≠"NUM" Then 
   Pause error_n&tp_req&"]" 
   Cycle 
  EndIf 
  0→cont 
  For i,1,5 
   l_ingad[i]→l_ingadi 
   l_ingadi&"_s"→l_ingads 
   If #l_ingads≠"" Then 
    cont+1→cont 
    l_ingadi→ingad 
    l_ingads→ingad_s 
   EndIf 
  EndFor 
  If cont=1 Then 
   Try 
    abs(expr(#ingad_s))→#ingad 
   Else 
    Pause error_d&ingad 
    Cycle 
   EndTry 
   If getType(#ingad)≠"NUM" Then 
    Pause error_n&ingad&"]" 
    Cycle 
   EndIf 



  ElseIf cont>1 Then 
   Pause "Error: Únicamente se admite un ingreso adicional (los demás en 
blanco)" 
   Cycle 
  EndIf 
  tp_uni[opc_2]→tp_unio2 
  If opc_1=1 Then 
   If opc_2=2 or opc_2=4 
    "°"&tp_unio2→tp_unio2 
   "tmpCnv("&tp_s&"_"&tp_unio2&",_"&tp_uni1&")"→tp_conv 
  Else 
   tp_s&"_"&tp_unio2&"▶_"&tp_uni1→tp_conv 
  EndIf 
  part(expr(tp_conv),1)→tp 
  If tp<expr(tp_lim[1]) or tp>expr(tp_lim[2]) 
   Pause "Advertencia: El valor de "&tp_req&" se encuentra fuera del 
intervalo de datos, se procederá con el valor más cercano disponible" 
  termo\tp_a89(opc_1+7,tp)→tpa89_ip 
  If cont=0 Then 
   
[["T,°C","T,°C","v',m³/kg","v'',m³/kg","ρ',kg/m³","ρ'',kg/m³","u',kJ/kg",
"u'',kJ/kg","h',kJ/kg","h'',kJ/kg","s',kJ/(kg*K)","s'',kJ/(kg*K)"]]→rta_1 
   newMat(1,12)→rta 
   tpa89_ip[1,1]→rta[1,1] 
   tpa89_ip[1,2]→rta[1,2] 
   1/(tpa89_ip[1,3])→rta[1,3] 
   tpa89_ip[1,4]→rta[1,4] 
   tpa89_ip[1,3]→rta[1,5] 
   1/(tpa89_ip[1,4])→rta[1,6] 
   tpa89_ip[1,5]-100*rta[1,3-opc_1]*rta[1,3]→rta[1,7] 
   tpa89_ip[1,6]-100*rta[1,3-opc_1]*rta[1,4]→rta[1,8] 
   tpa89_ip[1,5]→rta[1,9] 
   tpa89_ip[1,6]→rta[1,10] 
   tpa89_ip[1,7]→rta[1,11] 
   tpa89_ip[1,8]→rta[1,12] 
  Else 
   If ingad≠"x" Then 
    If ingad="v" Then 
     1/(tpa89_ip[1,3])→y_f 
     tpa89_ip[1,4]→y_g 
    ElseIf ingad="u" Then 
     tpa89_ip[1,5]-100*tpa89_ip[1,3-opc_1]/(tpa89_ip[1,3])→y_f 
     tpa89_ip[1,6]-100*tpa89_ip[1,3-opc_1]*tpa89_ip[1,4]→y_g 
    ElseIf ingad="h" Then 
     tpa89_ip[1,5]→y_f 
     tpa89_ip[1,6]→y_g 
    Else 
     tpa89_ip[1,7]→y_f 
     tpa89_ip[1,8]→y_g 



    EndIf 
    (#ingad-y_f)/(y_g-y_f)→x 
   EndIf 
   If x<0 or x>1 Then 
    Disp "La calidad x="&string(x) 
    If x<0 Then 
     Pause "Advertencia: Es líquido subenfriado, se calculá para líquido 
saturado (x=0)" 
     0→x 
    ElseIf x>1 Then 
     Pause "Advertencia: Es vapor sobrecalentado, se calculá para vapor 
saturado (x=1)" 
     1→x 
    EndIf 
   EndIf 
   
[["T,°C","T,°C","x","v,m³/kg","ρ,kg/m³","u,kJ/kg","h,kJ/kg","s,kJ/(kg*K)"
]]→rta_1 
   newMat(1,8)→rta 
   tpa89_ip[1,1]→rta[1,1] 
   tpa89_ip[1,2]→rta[1,2] 
   x→rta[1,3] 
   1/(tpa89_ip[1,3])→y_f 
   tpa89_ip[1,4]→y_g 
   (1-x)*y_f+x*y_g→rta[1,4] 
   1/(rta[1,4])→rta[1,5] 
   tpa89_ip[1,5]-100*tpa89_ip[1,3-opc_1]/(tpa89_ip[1,3])→y_f 
   tpa89_ip[1,6]-100*tpa89_ip[1,3-opc_1]*tpa89_ip[1,4]→y_g 
   (1-x)*y_f+x*y_g→rta[1,6] 
   tpa89_ip[1,5]→y_f 
   tpa89_ip[1,6]→y_g 
   (1-x)*y_f+x*y_g→rta[1,7] 
   tpa89_ip[1,7]→y_f 
   tpa89_ip[1,8]→y_g 
   (1-x)*y_f+x*y_g→rta[1,8] 
  EndIf 
  "p,bar"→rta_1[1,3-opc_1] 
  augment(rta_1;rta)→termo\rta_fm 
  Disp "▫termo\rta_fm" 
  Pause termo\rta_fm 
 EndLoop 
EndLoop 
setMode("Exact/Approx",m_ea) 
setMode("Display Digits",m_dd) 
DispHome 
EndPrgm 
 

 
 



tp_a89(a89,tp_i) 
Func 
© Fecha:   21-03-2008 
© Título:  Propiedades del Agua Saturada 
© Args.:   a89:  Número de la tabla a usar (a89≠9 => A.8 y a89=9 => A.9) 
©                A.8: Tabla de temperatura 
©                A.9: Tabla de presión 
©          tp_i: Temperatura [°C] ó presión [bar] de saturación (según 
'a89') 
© Versión: 1.0 
© Autor:   Froilán Andrés Morales Sanabria <froilan1986@hotmail.com> 
©          http://www.ticalc.org/archives/files/authors/95/9551.html 

Local s,n,tp,i,tp_1,tp_2,y_1,y_2,y_i 
"termo\tabla_a"→s 
If a89≠9 Then 
 s&"8"→s 
 75→n 
Else 
 s&"9"→s 
 67→n 
EndIf 
subMat(#s,2,1,n+1,1)→tp 
If tp_i≥tp[1,1] and tp_i≤tp[n,1] Then 
 For i,2,n 
  If tp_i≤tp[i,1] 
   Exit 
 EndFor 
 tp[i-1,1]→tp_1 
 tp[i,1]→tp_2 
 subMat(#s,i,1,i,8)→y_1 
 subMat(#s,i+1,1,i+1,8)→y_2 
 (y_2-y_1)/(tp_2-tp_1)*(tp_i-tp_1)+y_1→y_i 
ElseIf tp_i<tp[1,1] Then 
 subMat(#s,2,1,2,8)→y_i 
ElseIf tp_i>tp[n,1] Then 
 subMat(#s,n+1,1,n+1,8)→y_i 
EndIf 
Return y_i 
EndFunc 
 

 


